
Slides for Chapter 8: Open Economy Models for Competitive
Economies

Copyright: James R. Markusen University of Colorado, Boulder

8.1 Small open economy

Small open economy 2x2 (two goods, two factors) in which the rest of
the world is not explicitly modeled.  

Know as the “small-country assumption”: the country faces fixed
world prices

Trading opportunities are summarized by simple functions which
allow the economy to transform one good (an export) into another
(an import).  

The “technology” of these functions represents world prices.  We will
assume that these technologies or price ratios are fixed in this
model.  

Country exports  X1 in exchange for X2 (activities E1, M2)

We will allow for the fact that some policy or endowment change
could actually reverse the direction of trade by specifying (initially
inactive) functions that transform goods in the opposite direction.  

Initial data, in which 50 units of good  X1 are exchanged for 50 units
of good X2 at an implicit price ratio of p1/p2 = 1.  

          Production Sectors                      Consumer

Markets |     X1      X2      E1      M2      W       CONS
---------------------------------------------------------
P1      |     150            -50             -100
P2      |             50              50     -100
PL      |    -135     -5                               140
PK      |     -15    -45                                60
PW      |                                     200     -200
PFX     |                     50     -50
-----------------------------------------------------------

Technology parameters are specified in these functions that allow the
modeler to change the terms of trade.

PARAMETERS
   PE2     Export price of good 2,
   PM1     Import price of good 1,
   PE1     Export price of good 1,
   PM2     Import price of good 2,
   TM2     Import tariff for good 2;

PE1 = 1;
PM2 = 1;
PE2 = 0.999;
PM1 = 1.001;

E1 and M2 production activities that are the initially active trade links. 

Small trade costs on inactive links prevent model degeneracy and
“round tripping”.  

If all prices = 1, then if E1 = M2 = 1 is a solution, then so is 
E1 = 2, M1 = 1, M2 = 1.

While we could specify this activity as directly transforming X1 into X2
, in more complicated models it proves useful to define another
good which we will call “foreign exchange” and whose price is
denoted PFX.  

All trade is mediated through the “foreign exchange market”.

Thus activity E1 transforms X1 into foreign exchange and M2, the
import activity for good 2, transforms foreign exchange into
imports of good 2. 

There are four trade activities as just noted.

PRF_E1..  0*P1 =G= 50*PFX*PE1;

PRF_E2..  50*P2 =G= 50*PFX*PE2;

PRF_M1..  50*PFX*PM1 =G= 50*P1;

PRF_M2..  50*PFX*PM2 =G= 50*P2;

The equation for “foreign exchange is the trade balance condition:

MKT_PFX.. 50*E2*PE2 + 50*E1*PE1 =G= 50*PM2*M2 + 50*PM1*M1; 

$TITLE  M8-1.GMS: Small open economy 2x2
* strips out trade costs and tariffs for simplicity of exposition

$ONTEXT
 CALIBRATION: country exports X1, imports X2
 in free-trade SOE

                Production Sectors                      Consumer

Markets   |     X1      X2      E1      M2      W       CONS
----------------------------------------------------------------
P1        |     150            -50             -100
P2        |             50              50     -100
PL        |    -135     -5                               140
PK        |     -15    -45                                60
PW        |                                     200     -200
PFX       |                     50     -50
----------------------------------------------------------------
$OFFTEXT

* the first four parameters allow changes in (exogenous) world prices

PARAMETERS
        P E 2     Export price of good 2          /0.999/
        P M 1     Import price of good 1          /1.001/

        P E 1     Export price of good 1          /1/
        P M 2     Import price of good 2          /1/;

NONNEGATIVE VARIABLES
        X 1      Activity level for sector X1
        X 2      Activity level for sector X2
        E 1      Activity level for sector E1
        E 2      Activity level for sector E2
        M 1      Activity level for sector M1
        M 2      Activity level for sector M2
        W       Activity level for sector W

        P 1      Price index for commodity X
        P 2      Price index for commodity Y
        P L      Price index for primary factor L
        P K      Price index for primary factor K
        P W      Price index for welfare (consumer price index)
        P F X     Real exchange rate index

        C O N S    Income definition for CONS;

EQUATIONS
        P R F _ X 1  Zero profit for sector X1
        P R F _ X 2  Zero profit for sector X2
        P R F _ E 1  Zero profit for sector E1
        P R F _ E 2  Zero profit for sector E2

        P R F _ M 1  Zero profit for sector M1
        P R F _ M 2  Zero profit for sector M2
        P R F _ W   Zero profit for sector W

        M K T _ X 1  Supply-demand balance for commodity X1
        M K T _ X 2  Supply-demand balance for commodity X2
        M K T _ P F X Supply-demand balance for commodity PFX
        M K T _ L   Supply-demand balance for primary factor L
        M K T _ K   Supply-demand balance for primary factor L
        M K T _ W   Supply-demand balance for aggregate demand

        I _ C O N S  Income definition for CONS;

*       Zero profit conditions

PRF_X1..    150*PL**(0.9) * PK**(0.1) =G= 150*P1;

PRF_X2..    50*PL**(0.1) * PK**(0.9) =G=  50*P2;

PRF_E1..    50*P1 =G= 50*PFX*PE1;

PRF_E2..    50*P2 =G= 50*PFX*PE2;

PRF_M1..    50*PFX*PM1 =G= 50*P1;

PRF_M2..    50*PFX*PM2 =G= 50*P2;

PRF_W..     100*P1**0.5 * P2**0.5 =G= 100*PW;

*       Market clearance conditions

MKT_X1..    150*X1 + 50*M1 =G= 50*E1 + 100*W*PW/P1;

MKT_X2..    50*X2  + 50*M2 =G= 50*E2 + 100*W*PW/P2 ;

MKT_PFX..   50*E2*PE2 + 50*E1*PE1 =G= 50*PM2*M2 + 50*PM1*M1;

MKT_W..     200*W =G= CONS / PW;

MKT_L..     140 =G= 135*X1 * P1/PL +  5*X2*P2/PL;

MKT_K..     60 =G=   15*X1*P1/PK + 45*X2*P2/PK;

*       Income balance

I_CONS..    CONS =E= 140*PL + 60*PK;

MODEL SOE1 /PRF_X1.X1, PRF_X2.X2, PRF_E1.E1, PRF_E2.E2,
                 PRF_M1.M1, PRF_M2.M2, PRF_W.W,
                 MKT_X1.P1, MKT_X2.P2, MKT_PFX.PFX, MKT_L.PL,
                 MKT_K.PK, MKT_W.PW, I_CONS.CONS /;

*       set SOE values:

X1.L    =1;
X2.L    =1;
E2.L    =0;
M1.L    =0;
E1.L    =1;
M2.L    =1;
W.L     =1;

P1.L    =1;
P2.L    =1;
PFX.L   =1;
PK.L    =1;
PL.L    =1;
CONS.L  =200;

* choose the real consumer price index as numeraire

PW.FX   =1;

* check for calibration and starting-value errors

SOE1.ITERLIM = 0;
SOLVE SOE1 USING MCP;
SOE1.ITERLIM = 2000;

SOLVE SOE1 USING MCP;

* counterfactual: a terms-of-trade improvement

PE1 = 1.2;
PM1 = 1.21;
SOE1.ITERLIM = 2000;
SOLVE SOE1 USING MCP;

8.2 Small open economy: tariffs versus trade costs

Zero-profit conditions for trade activities

PFX - price of a unit of “foreign exchange”
TC - gross real trade cost: 1 + the trade-cost rate
TR - gross tariff: 1 + the ad valorem tariff rate

TC = TR = 1: costless trade:    tariffs, trade costs appear symmetric

PRF_E1..    50 * P1 =G= 50 * PFX * PE1;

PRF_E2..    50 * P2 =G= 50 * PFX * PE2;

PRF_M1..    50 * PFX * PM1 * TC * TR =G= 50 * P1;

PRF_M2..    50 * PFX * PM2 * TC * TR =G= 50 * P2;

TC and TR appear the same way in the pricing equations.  However,
they are quite different from one another.

Modeling a tariff that is costlessly collected and redistributed to
consumers lump sum.

Recall that TR is defined as the gross tariff (one plus the ad valorem
rate), so (TR - 1) is the tariff rate.  The tariff revenue from
equation PRF_M1 is:

[50 * PFX * PM2] * (TR-1) * M2

The term in brackets [ ] is the cost of 50 units (the benchmark value)
of imports of good 2, (TR - 1) is the tariff rate, and M2 is the
activity level, calibrated at one initially.  

The assumption that the tariff can be costlessly redistributed means
that this amount must be entered in the consumer’s budget
constraint: Income = factor income + tariff revenue.

CONS =E= 140*PL + 60*PK + 50*PFX*PM2*(TR-1)*M2;   

Modeling an “iceberg” trade cost.  Assume that the trade cost is paid
in units of the good itself.  

This is sometimes referred to as “iceberg trade costs”.  Some of the
good “melts” in transit and less arrives than is shipped.  

Let p* and M be the price received by the exporter and M be the
number of units shipped.  

Then the price paid by the importer is (p*TC).  The quantity received
by the importer is (M/TC): the quantity that arrives unmelted.

Note that revenue received by the exporter and paid by the importer
are the same: the payments do and must balance.

p*M = (p*TC)(M/TC)

The trade cost obviously does not generate any revenue and does
not become a term in the consumer’s budget constraint.

However, it reappears in the model in the supply = demand equations
for the two goods: the amount entered is the amount received, not
the amount shipped.

The market-clearing conditions for the two goods are production plus
imports (received) minus exports equal domestic demand.

150*X1 + 50*M1/TC  =G=  50*E1 + 100*W*PW/P1

 50*X2 + 50*M2/TC  =G=  50*E2 + 100*W*PW/P2

Finally, there is a market-clearing condition for foreign exchange that
determines it price (PFX): 

earnings from exports = import costs

50*E2*PE2 + 50*E1*PE1 =G= 50*PM2*M2 + 50*PM1*M1;

Counter-factual: show the welfare effects of a tariff versus a real
trade costs.

Two sets are defined: 

I will loop over values of trade costs or tariffs
J will denote the tariff and trade cost scenarios respectively.  

Then some two-dimensional parameters are defined in order to
extract some results after solving.  

The * in RESULTS(I, *) is called a “wild card” and its use should be
obvious below.    

SETS I indexes 25 different gross cost levels      /I1*I25/
     J indexes 2 scenarios: 1 = tar 2 = trade cost /J1*J2/;

PARAMETERS
 RATE(I)
 WELFARE(I,J)

  IMP2(I,J)
 TRATE(I,J)
 TCOST(I,J)
 RESULTS(I, *);

Then there is the double loop, the inner loop over I sets the cost rate,
starting at 1. 

For J = 1, the rate is a tariff, for J = 2, the rate is a trade costs.  

LOOP(J,
LOOP(I,

TC = 1; TR = 1;
RATE(I) = 1 + .05*ORD(I) - 0.05;
TR$(ORD(J) EQ 1) = RATE(I);
TC$(ORD(J) EQ 2) = RATE(I);

The left-hand exception operator $ 

 TR$(ORD(J) EQ 1) = RATE(I);

is read “set the parameter TR equal to RATE if the ordinal value of J
is equal to one, otherwise leave TR unchanged from it’s existing
value”. 

For future reference, the meaning is different if the exception operator
is on the right-hand side.  

TR= RATE(I)$(ORD(J) EQ 1;

is read “set the tariff TR equal to RATE if the ordinal value of J is
equal to one, otherwise set TR equal to zero”.  This is something
quite different!  

The parameter IMP(I, J) extracts the equilibrium value of exports in
the two scenarios.

SOLVE ALGEBRAIC USING MCP;

WELFARE(I,J) = W.L;
IMP2(I,J) = M2.L/TC;

);
);

RESULTS(I, "RATE") = RATE(I);
RESULTS(I, "WELTR") = WELFARE(I, "J1");
RESULTS(I, "WELTC") = WELFARE(I, "J2");
RESULTS(I, "IMP2TR")= IMP2(I,"J1");
RESULTS(I, "IMP2TC")= IMP2(I,"J2");

DISPLAY RESULTS;

$TITLE  M8-2.GMS: Small open economy 2x2
*       contrasts tariffs versus (iceberg) trade costs

$ONTEXT
 CALIBRATION: country exports X1, imports X2
 in free-trade benchmark

                Production Sectors                      Consumer

Markets   |     X1      X2      E1      M2      W       CONS
----------------------------------------------------------------
P1        |     150            -50             -100
P2        |             50              50     -100
PL        |    -135     -5                               140
PK        |     -15    -45                                60
PW        |                                     200     -200
PFX       |                     50     -50
----------------------------------------------------------------
$OFFTEXT

* the first four parameters allow changes in (exogenous) world

PARAMETERS
        P E 2     Export price of good 2          /0.999/
        P M 1     Import price of good 1          /1.001/

        P E 1     Export price of good 1          /1/
        P M 2     Import price of good 2          /1/
        T R      Import tariff (gross basis)     /1/
        T C      Trade costs (gross basis)       /1/
        C O N S X 1  Consumption of X1
        C O N S X 2  Consumption of X2;

NONNEGATIVE VARIABLES
        X 1      Activity level for sector X1
        X 2      Activity level for sector X2
        E 1      Activity level for sector E1
        E 2      Activity level for sector E2
        M 1      Activity level for sector M1
        M 2      Activity level for sector M2
        W       Activity level for sector W

        P 1      Price index for commodity X
        P 2      Price index for commodity Y
        P L      Price index for primary factor L
        P K      Price index for primary factor K
        P W      Price index for welfare (consumer price index)
        P F X     Real exchange rate index

        C O N S    Income definition for CONS;

EQUATIONS
        P R F _ X 1  Zero profit for sector X1
        P R F _ X 2  Zero profit for sector X2
        P R F _ E 1  Zero profit for sector E1
        P R F _ E 2  Zero profit for sector E2
        P R F _ M 1  Zero profit for sector M1
        P R F _ M 2  Zero profit for sector M2
        P R F _ W   Zero profit for sector W

        M K T _ X 1  Supply-demand balance for commodity X1
        M K T _ X 2  Supply-demand balance for commodity X2
        M K T _ P F X Supply-demand balance for commodity PFX
        M K T _ L   Supply-demand balance for primary factor L
        M K T _ K   Supply-demand balance for primary factor L
        M K T _ W   Supply-demand balance for aggregate demand

        I _ C O N S  Income definition for CONS;

*       Zero profit conditions

PRF_X1..    150*PL**(0.9) * PK**(0.1) =G= 150*P1;

PRF_X2..    50*PL**(0.1) * PK**(0.9) =G=  50*P2;

PRF_E1..    50*P1 =G= 50*PFX*PE1;

PRF_E2..    50*P2 =G= 50*PFX*PE2;

PRF_M1..    50*PFX*PM1*TC*TR =G= 50*P1;

PRF_M2..    50*PFX*PM2*TC*TR =G= 50*P2;

PRF_W..     100*P1**0.5 * P2**0.5 =G= 100*PW;

*       Market clearance conditions

MKT_X1..    150*X1 + 50*M1/TC =G=  50*E1 + 100*W*PW/P1;

MKT_X2..    50*X2 + 50*M2/TC =G= 50*E2 + 100*W*PW/P2 ;

MKT_PFX..   50*E2*PE2 + 50*E1*PE1 =G= 50*PM2*M2 + 50*PM1*M1;

MKT_W..     200*W =G= CONS / PW;

MKT_L..     140 =G= 135*X1 * P1/PL +  5*X2*P2/PL;

MKT_K..     60 =G=   15*X1*P1/PK + 45*X2*P2/PK;

*       Income balance

I_CONS..    CONS =E= 140*PL + 60*PK + 50*PFX*PM2*(TR-1)* M2;

MODEL SOE /PRF_X1.X1, PRF_X2.X2, PRF_E1.E1, PRF_E2.E2,
                 PRF_M1.M1, PRF_M2.M2, PRF_W.W,
                 MKT_X1.P1, MKT_X2.P2, MKT_PFX.PFX, MKT_L.PL,
                 MKT_K.PK, MKT_W.PW, I_CONS.CONS /;

*       set benchmark values:

X1.L    =1;
X2.L    =1;
E2.L    =0;
M1.L    =0;
E1.L    =1;
M2.L    =1;
W.L     =1;

P1.L    =1;
P2.L    =1;
PFX.L   =1;
PK.L    =1;
PL.L    =1;
CONS.L  =200;

* choose the real consumer price index as numeraire

PW.FX   =1;

* check for calibration and starting-value errors

SOE.ITERLIM = 0;
SOLVE SOE USING MCP;

SOE.ITERLIM = 2000;
SOLVE SOE USING MCP;

* SHOW HOW TO DO MULTIPLE SCENARIOS
* SHOW DIFFERENCE BETWEEN TARIFF AND TRADE COST OF EQUAL RATES

SETS I indexes 25 different gross cost levels         /I1*I25/
     J indexes 2 scenarios: 1 = tariff 2 = trade cost /J1*J2/;

PARAMETERS
 RATE(I)
 WELFARE(I,J)
 IMP2(I,J)
 TRATE(I,J)
 TCOST(I,J)
 RESULTS(I, *);

LOOP(J,
LOOP(I,

TC = 1; TR = 1;
RATE(I) = 1 + .05*ORD(I) - 0.05;
TR$(ORD(J) EQ 1) = RATE(I);
TC$(ORD(J) EQ 2) = RATE(I);

SOLVE SOE USING MCP;

WELFARE(I,J) = W.L;
IMP2(I,J) = M2.L/TC;

);
);

RESULTS(I, "RATE") = RATE(I);
RESULTS(I, "WELTR") = WELFARE(I, "J1");
RESULTS(I, "WELTC") = WELFARE(I, "J2");
RESULTS(I, "IMP2TR")= IMP2(I,"J1");
RESULTS(I, "IMP2TC")= IMP2(I,"J2");

DISPLAY RESULTS;

* Write parameter RESULTS to an Excel file M8.XLS,
* starting in Sheet1

Execute_Unload 'M8.gdx' RESULTS
execute 'gdxxrw.exe M8.gdx par=RESULTS rng=SHEET1!'

* simpler but less sophisticated dump to excel

$LIBINCLUDE XLDUMP RESULTS M8.XLS SHEET2!

8.3 Small open economy: calibrating to tariffs in the benchmark

This model has a 20% tariff in the benchmark data.  It is important to
keep track of prices and trade balance in this situation.

          Production Sectors                 Consumer
Markets|  X1      X2      E1      M2      W      CONS
---------------------------------------------------- 
P1     | 150             -50            -100
P2     |          40              60    -100
PL     |-100     -20                              120
PK     | -50     -20                               70
PW     |                                 200     -200
PFX    |                  50     -50             
Tariff |                         -10               10
-----------------------------------------------------

Trade balance holds, since exports of good 1 generate 50 units of
foreign exchange and those 50 units are spent on imports. 

Consumers spend 60 on imports of which 10 is the tariff, or 20%
of the value of imports (10/50). 

The convention we adopt here is that all domestic prices are equal to
1 initially.  Since the export good X1 is freely traded, then the
international price of X1 is also equal to 1.  

However, the import good has a domestic price equal to (1 + TM2)
times the world price, so if the domestic price is equal to 1 then
the world price must equal 1/ (1 + TM2) = 1/1.2.

Thus the -50 in the column M2 of the matrix is interpreted as 60 units
at a price of 1/1.2. 

The counterfactual experiment is free trade.

$TITLE: M8-3: Small open economy with a benchmark tariff

$ONTEXT
In this example, units are chosen such that all DOMESTIC prices
equal one initially.  Implied world price of import good
X2: P2 = 1/1.2

                Production Sectors                     Consumer

Markets       |     X1      X2      E1      M2      W      CONS
---------------------------------------------------------------
P1            |    150             -50            -100
P2            |             40              60    -100
PL            |   -100     -20                              120
PK            |    -50     -20                               70
PW            |                                    200     -200
PFX           |                     50     -50
TARIFF        |                            -10               10
---------------------------------------------------------------
$OFFTEXT

PARAMETERS
        P E 2     Export price of good 2
        P M 1     Import price of good 1
        P E 1     Export price of good 1
        P M 2     Import price of good 2

        T M 2     Import tariff for good 2;

PE1     = 1;
PM2     = 1 / (1.2);
PE2     = PM2 * 0.99;
PM1     = 1.01;
TM2     = 0.2;

POSITIVE VARIABLES
        X 1      Activity level for sector X1
        X 2      Activity level for sector X2
        E 1      Activity level for sector E1
        E 2      Activity level for sector E2
        M 1      Activity level for sector M1
        M 2      Activity level for sector M2
        W       Activity level for sector W (Hicksian welfare index)
        P 1      Price index for commodity X
        P 2      Price index for commodity Y
        P L      Price index for primary factor L
        P K      Price index for primary factor K
        P W      Price index for welfare (expenditure function)
        P F X     Read exchange rate index
        C O N S    Income definition for CONS;

EQUATIONS
        P R F _ X 1  Zero profit for sector X1
        P R F _ X 2  Zero profit for sector X2
        P R F _ E 1  Zero profit for sector E1
        P R F _ E 2  Zero profit for sector E2
        P R F _ M 1  Zero profit for sector M1
        P R F _ M 2  Zero profit for sector M2
        P R F _ W   Zero profit for sector W (Hicksian welfare index)

        M K T _ X 1  Supply-demand balance for commodity X1
        M K T _ X 2  Supply-demand balance for commodity X2
        M K T _ P F X Supply-demand balance for commodity PFX
        M K T _ L   Supply-demand balance for primary factor L
        M K T _ K   Supply-demand balance for primary factor L
        M K T _ W   Supply-demand balance for aggregate demand

        I _ C O N S  Income definition for CONS;

*       Zero profit conditions

PRF_X1..        150 * PL**(2/3) * PK**(1/3) =G= 150 * P1;

PRF_X2..        40  * PL**(0.5) * PK**(0.5) =G= 40  * P2;

PRF_E1..        50 * P1 =G= 50 * PFX * PE1;

PRF_E2..        60 * P2 =G= 60 * PFX * PE2;

PRF_M1..        50 * PFX * PM1 =G= 50 * P1;

PRF_M2..        60 * PFX * PM2 *(1+TM2) =G= 60 * P2;

PRF_W..         200 * P1**0.5 * P2**0.5 =G= 200 * PW;

*       Market clearance conditions

MKT_X1..        150 * X1 + 50 * M1 =G=  50 * E1 + 100 * W *PW/P1;

MKT_X2..        40 * X2 + 60 * M2 =G= 60 * E2 + 100 *W *PW/P2 ;

MKT_PFX..       60 * E2 * PE2 + 50 * E1 * PE1 =G=
                60 * M2 * PM2 + 50 * PM1 * M1;

MKT_W..         200 * W =G= CONS / PW;

MKT_L..         120  =G= 100 * X1 * P1/PL + 20 * X2 * P2/PL;

MKT_K..         70 =G= 50 * X1 * P1/PK + 20 * X2 * P2/PK;

*       Income balance

I_CONS..        CONS =E= 120*PL + 70*PK + 60 * PFX * PM2 * M2 * TM2;

MODEL SOETARIFF /PRF_X1.X1, PRF_X2.X2, PRF_E1.E1, PRF_E2.E2,
                 PRF_M1.M1, PRF_M2.M2, PRF_W.W,
                 MKT_X1.P1, MKT_X2.P2, MKT_PFX.PFX, MKT_L.PL,
                 MKT_K.PK, MKT_W.PW, I_CONS.CONS /;

*       Check the benchmark (again):

X1.L    =1;
X2.L    =1;
E2.L    =0;
M1.L    =0;
E1.L    =1;
M2.L    =1;
W.L     =1;

P1.L    =1;
P2.L    =1;
PFX.L   =1;
PK.L    =1;
PW.FX   =1;
PL.L    =1;

CONS.L  =200;

SOETARIFF.ITERLIM = 0;

SOLVE SOETARIFF USING MCP;

SOETARIFF.ITERLIM = 2000;
SOLVE SOETARIFF USING MCP;

* ccounterfactual experiment: free trade

TM2  = 0 ;
SOLVE SOETARIFF USING MCP;

8.4a Small open economy: modeling a quota as an endogenous
tax equivalent

This model assumes the same benchmark data as the previous two
models, but it assumes that there is a quota limiting imports.

The quota generates a gap between the foreign or world supply price
and the domestic demand price for the “rationed” good.  

This difference, often referred to as a quota rent (it is a form of
Ricardian rent), must go to some agent.

In this model, we model this as a “tariff equivalent”.  

The size of the tariff equivalent is endogenous, set so that the level
of imports is less than or equal to the quota level. 

We use an auxiliary variable, PQ,  for an endogenous tax rate and a
constraint equation to set the value of this “tax”.  

The revenue from this tax is assigned to the representative
consumer in the block for the import demand for M2.  

The constraint equation (A_PQ) says to set the value of the tax PQ
such that the activity level for imports (calibrated to be 1 initially)
is less than or equal to 1.

A_PQ..    1 =G= M2;    

And “tax” revenue (quota rent) is returned to the consumer.

CONS =E= 120*ENDOW*PL + 70*ENDOW*PK
                 + 60*PFX*PM2*M2*PQ;

Units are chosen such that all domestic prices equal 1.  The
counterfactual experiment is to set remove the quota.  

This is done with the statement PQ.FX = 0; since PQ is a variable,
not a parameter.  

An alternative way to do this is to declare a parameter and use it in
place of the ‘1’ on the left-hand side of the constraint equation. 
Then the quota can be set at any level.  

The value of PQ can be thought of as the tariff equivalent of the
quota or the “shadow tariff”.  To reintroduce the quota, we must
free up PQ with two statements:

PQ.L = 0;
PQ.UP = +INF;

The second counterfactual is to triple the size of the economy.

$TITLE  M8-4a.GMS: Small open economy with a benchmark quota
* modeled as an endogenous tax rate

$ONTEXT
In this example, units are chosen such that all DOMESTIC prices
equal one initially.  Implied world price of import good
X2: P2 = 1/1.2

                Production Sectors                     Consumer

Markets       |     X1      X2      E1      M2      W      CONS
---------------------------------------------------------------
P1            |    150             -50            -100
P2            |             40              60    -100
PL            |   -100     -20                              120
PK            |    -50     -20                               70
PW            |                                    200     -200
PFX           |                     50     -50
Q (quota rent)|                            -10               10
---------------------------------------------------------------
$OFFTEXT

PARAMETERS
        P E 2     Export price of good 2
        P M 1     Import price of good 1
        P E 1     Export price of good 1

        P M 2     Import price of good 2
        E N D O W   Endowment multiplier (size of the economy);

PE1     = 1;
PM2     = 1 / (1.2);
PE2     = PM2 * 0.99;
PM1     = 1.01;
ENDOW   = 1;

NONNEGATIVE VARIABLES
    X 1      Activity level for sector X1,
    X 2      Activity level for sector X2,
    E 1      Activity level for sector E1,
    E 2      Activity level for sector E2,
    M 1      Activity level for sector M1,
    M 2      Activity level for sector M2,
    W       Activity level for sector W,
    P 1      Price index for commodity X,
    P 2      Price index for commodity Y,
    P L      Price index for primary factor L,
    P K      Price index for primary factor K,
    P W      Price index for welfare (expenditure function),
    P F X     Read exchange rate index,
    C O N S    Income definition for CONS
    P Q      Quota shadow price (ad valorem tariff equivalent);

EQUATIONS
    P R F _ X 1  Zero profit for sector X1
    P R F _ X 2  Zero profit for sector X2
    P R F _ E 1  Zero profit for sector E1
    P R F _ E 2  Zero profit for sector E2
    P R F _ M 1  Zero profit for sector M1
    P R F _ M 2  Zero profit for sector M2
    P R F _ W   Zero profit for sector W

    M K T _ X 1  Supply-demand balance for commodity X1
    M K T _ X 2  Supply-demand balance for commodity X2
    M K T _ P F X Supply-demand balance for commodity PFX
    M K T _ L   Supply-demand balance for primary factor L
    M K T _ K   Supply-demand balance for primary factor L
    M K T _ W   Supply-demand balance for aggregate demand

    I _ C O N S  Income definition for CONS
    A _ P Q    Quota auxiliary (sets endogenous shadow tax PQ);

*       Zero profit conditions

PRF_X1..  150 * PL**(2/3) * PK**(1/3) =G= 150*P1;

PRF_X2..  40  * PL**(0.5) * PK**(0.5) =G= 40*P2;

PRF_E1..  50 * P1 =G= 50 * PFX * PE1;

PRF_E2..  60 * P2 =G= 60 * PFX * PE2;

PRF_M1..  50 * PFX * PM1 =G= 50 * P1;

PRF_M2..  60 * PFX * PM2 *(1+PQ) =G= 60*P2;

PRF_W..   200 * P1**0.5 * P2**0.5 =G= 200*PW;

*       Market clearance conditions

MKT_X1..  150*X1 + 50*M1 =G=  50*E1 + 100*W*PW/P1;

MKT_X2..  40*X2 + 60*M2 =G= 60*E2 + 100*W*PW/P2 ;

MKT_PFX.. 60*E2*PE2 + 50*E1*PE1 =G= 60*M2*PM2 + 50*PM1*M1;

MKT_W..   200*W =G= CONS / PW;

MKT_L..   120*ENDOW  =G= 100*X1*P1/PL + 20*X2*P2/PL;

MKT_K..   70*ENDOW   =G=  50*X1*P1/PK + 20*X2*P2/PK;

*       Income balance

I_CONS..  CONS =E= 120*ENDOW*PL + 70*ENDOW*PK + 60*PFX*PM2*M2*PQ;

A_PQ..    1 =G= M2;

MODEL SOE2 /PRF_X1.X1, PRF_X2.X2, PRF_E1.E1, PRF_E2.E2,
                 PRF_M1.M1, PRF_M2.M2,
                 PRF_W.W, MKT_X1.P1, MKT_X2.P2, MKT_PFX.PFX,
                 MKT_L.PL, MKT_K.PK, MKT_W.PW,
                 I_CONS.CONS, A_PQ.PQ /;

*       Check the benchmark (again):

X1.L    =1;
X2.L    =1;
E2.L    =0;
M1.L    =0;
E1.L    =1;
M2.L    =1;
W.L     =1;

P1.L    =1;
P2.L    =1;
PFX.L   =1;
PK.L    =1;
PW.FX   =1;
PL.L    =1;

CONS.L  =200;

PQ.L = 0.20;

SOE2.ITERLIM = 0;
SOLVE SOE2 USING MCP;

SOE2.ITERLIM = 2000;
SOLVE SOE2 USING MCP;

* counterfactual: fixed PQ = 0 to calculate free trade

PQ.FX = 0;
SOLVE SOE2 USING MCP;

* show that the quota becomes more restrictive
* as the economy grows

PQ.LO = 0;
PQ.UP = +INF;

ENDOW = 3;
SOLVE SOE2 USING MCP;

* show what would happen if there had been a fixed 0.20 tariff
* instead of the quota

PQ.FX = 0.20;
SOLVE SOE2 USING MCP;

PQ.LO = 0;
PQ.UP = +INF;

* show what would happen if the economy were smaller than the benchmark

ENDOW = 0.25;
SOLVE SOE2 USING MCP;

8.4b Small open economy: modeling a quota as a supply/demand
for licenses

This is the same exercise as the previous M8.4a, except is shows an
alternative way of modeling a quota or indeed any quantitative
restriction.

This is to assume that the quota is enforced by licences sold to
importers.  

The government might auction them off, for example (which does
happen in practice for certain licenses and permits).

We will assume that the government returns the receipt lump sum to
the representative consumer.

Now, instead of an auxiliary variable and constraint equation, we
have an added variable PLIC, the price of a license, 

and a market clearing equation MKT_LIC which equates the
exogenous supply of licences (SLIC) to the demand for them.

PARAMETER
   SLIC   Supply of import licenses for X2;   

NON-NEGATIVE VARIABLE
   PLIC    Price of a license; 

EQUATION
   MKT_LIC Market for import licenses;                    

PRF_M2..  60*PLIC + 60*PFX*PM2 =G= 60*P2;
              
MKT_LIC.. 60*SLIC =G= 60*M2;    

I_CONS..  CONS =E= 120*ENDOW*PL + 70*ENDOW*PK + 60*PLIC*SLIC;

$TITLE  M8-4b: Small open economy with a benchmark quota
* modeled as supply/demand for import licenses

$ONTEXT
In this example, units are chosen such that all DOMESTIC
 prices equal one initially.
Implied world price of import good X2 P2 = 1/1.2

                Production Sectors                     Consumer

Markets       |     X1      X2      E1      M2      W      CONS
---------------------------------------------------------------
P1            |    150             -50            -100
P2            |             40              60    -100
PL            |   -100     -20                              120
PK            |    -50     -20                               70
PW            |                                    200     -200
PFX           |                     50     -50
PLIC          |                            -10               10
---------------------------------------------------------------

$OFFTEXT

PARAMETERS
        P E 2     Export price of good 2

        P M 1     Import price of good 1
        P E 1     Export price of good 1
        P M 2     Import price of good 2
        S L I C    Supply of import licenses for X2 (M2)
        E N D O W   Endowment multiplier (size of the economy);

PE1     = 1;
PM2     = 1 / (1.2);
PE2     = PM2 * 0.99;
PM1     = 1.01;
SLIC    = 1;
ENDOW   = 1;

NONNEGATIVE VARIABLES
    X 1      Activity level for sector X1,
    X 2      Activity level for sector X2,
    E 1      Activity level for sector E1,
    E 2      Activity level for sector E2,
    M 1      Activity level for sector M1,
    M 2      Activity level for sector M2,
    W       Activity level for sector W,
    P 1      Price index for commodity X,
    P 2      Price index for commodity Y,
    P L      Price index for primary factor L,
    P K      Price index for primary factor K,
    P W      Price index for welfare (expenditure function),

    P F X     Read exchange rate index,
    C O N S    Income definition for CONS
    P L I C    Price of a license (ad valorem tariff equivalent);

EQUATIONS
    P R F _ X 1  Zero profit for sector X1
    P R F _ X 2  Zero profit for sector X2
    P R F _ E 1  Zero profit for sector E1
    P R F _ E 2  Zero profit for sector E2
    P R F _ M 1  Zero profit for sector M1
    P R F _ M 2  Zero profit for sector M2
    P R F _ W   Zero profit for sector W

    M K T _ X 1  Supply-demand balance for commodity X1
    M K T _ X 2  Supply-demand balance for commodity X2
    M K T _ P F X Supply-demand balance for commodity PFX
    M K T _ L   Supply-demand balance for primary factor L
    M K T _ K   Supply-demand balance for primary factor L
    M K T _ W   Supply-demand balance for aggregate demand

    I _ C O N S  Income definition for CONS
    M K T _ L I C Market for import licenses;

*       Zero profit conditions

PRF_X1..    150* PL**(2/3) * PK**(1/3) =G= 150*P1;

PRF_X2..    40* PL**(0.5) * PK**(0.5) =G= 40*P2;

PRF_E1..    50*P1 =G= 50*PFX*PE1;

PRF_E2..    60*P2 =G= 60*PFX*PE2;

PRF_M1..    50*PFX * PM1 =G= 50*P1;

PRF_M2..    60*PLIC + 60*PFX*PM2 =G= 60*P2;

PRF_W..     200*P1**0.5 * P2**0.5 =G= 200*PW;

*       Market clearance conditions

MKT_X1..    150*X1 + 50*M1 =G=  50*E1 + 100*W *PW/P1;

MKT_X2..    40*X2 + 60*M2 =G= 60*E2 + 100*W*PW/P2 ;

MKT_PFX..   60*E2*PE2 + 50*E1*PE1 =G= 60*M2*PM2 + 50*PM1*M1;

MKT_W..     200*W =G= CONS / PW;

MKT_L..     120*ENDOW  =G= 100*X1*P1/PL + 20*X2*P2/PL;

MKT_K..     70*ENDOW   =G=  50*X1*P1/PK + 20*X2*P2/PK;

MKT_LIC..   60*SLIC =G= 60*M2;

*       Income balance

I_CONS..    CONS =E= 120*ENDOW*PL + 70*ENDOW*PK + 60*PLIC*SLIC;

MODEL ALGEBRAIC /PRF_X1.X1, PRF_X2.X2, PRF_E1.E1, PRF_E2.E2,
                 PRF_M1.M1, PRF_M2.M2,
                 PRF_W.W, MKT_X1.P1, MKT_X2.P2, MKT_PFX.PFX,
                 MKT_L.PL, MKT_K.PK, MKT_W.PW,
                 MKT_LIC.PLIC, I_CONS.CONS/;

*       Check the benchmark:

X1.L    =1;
X2.L    =1;
E2.L    =0;
M1.L    =0;
E1.L    =1;
M2.L    =1;
W.L     =1;

P1.L    =1;
P2.L    =1;

PFX.L   =1;
PK.L    =1;
PW.FX   =1;
PL.L    =1;

CONS.L  =200;

PLIC.L = 1/6;

ALGEBRAIC.ITERLIM = 0;
SOLVE ALGEBRAIC USING MCP;

ALGEBRAIC.ITERLIM = 2000;
SOLVE ALGEBRAIC USING MCP;

* show what happens if the supply of licenses is greatly expanded

SLIC = 5;
SOLVE ALGEBRAIC USING MCP;

* show that the quota becomes more restrictive as
* the economy grows

SLIC = 1;
ENDOW = 3;
SOLVE ALGEBRAIC USING MCP;

8.5 Large economy and the optimal tariff 

For many countries, or for a small number of commodities for one
country, the small-country assumption of fixed world prices may
not be appropriate.  

One alternative is to model the whole world.  

But it may be convenient and appropriate for the modeler to stick
with the country and question and continue to represent the rest
of the world with trade transformation functions.  

We show how to do this, allowing for prices to change with quantities
supplied and demanded.

We declare and auxiliary variable, TOT for terms of trade, the
relative price of the country’s export good (good 1).  

Then there is a constraint equation T_TOT, which specifies that the
TOT falls as the country exports more of good 1. 

T_TOT  =G= E1**(-0.3) 

The TOT will also appear two other places in the model: in the
pricing equation for exports (the higher the TOT, the more foreign
exchange (FX) is earned from a unit of good one exported (E1)

PRF_E1..   50*P1 =G= 50*PFX*TOT;  

TOT also appears in the equation for trade balance which
determines the price of foreign exchange.

MKT_PFX.. 50*0.99*E2 + (50*TOT)*E1 =G= 50*M2 + 50*1.01*M1;  

Assumed there are many consumers/producers inside the
country: each views themselves as a price taker even though
the country as a whole has influence over world prices.

               Production Sectors                Consumer

Markets  |   X1      X2      E1      M2      W    CONS
-----------------------------------------------------------
P1       |  150             -50           -100
P2       |           50              50   -100
PL       | -100     -20                            120
PK       |  -50     -30                             80
PW       |                                 200    -200
PFX      |                   50     -50
-----------------------------------------------------------

We have specified this as a Cobb-Douglas function, so we are
assuming that it is of the form:

The elasticity of the relative export price with respect to export
quantity is -0.3.  Think of export earnings as:   

Then if you do the algebra, the elasticity of export earnings with
respect to export quantity is 0.7.  This is a strong term-of-trade
effect, and will lead to the usual optimal tariff argument for
protection.

Counterfactual is to impose a 0.20 tariff, which improves welfare.

Optimal tariff solved as an MPEC

First, add one variable (which is just welfare) and one equation.

VARIABLES
 WELOPT;

EQUATIONS
 WELFOPT;

WELFOPT..  WELOPT =E= W;

ADD the objective function to the model and DELETE the dummy
equation TARIFF.  

The tariff TM2 now does not appear as a matched variable but will
be solved for in the optimization.

$TITLE:  M8-5.GMS: Large open economy model
* modeled as an MCP and then as an MPEC to solve for the optimal tariff

$ONTEXT
This is similar to model M8-1.gms, but the terms of trade
now depend on the level of exports: high exports, lower prices
here this is modeled with an auxiliary variable TOT
where TOT is the relative price of exports to imports
TOT is set by the constraint equation T_TOT:  TOT = E1**(-0.3)

                Production Sectors                   Consumer

Markets  |   X1      X2      E1      M2      W    CONSH
----------------------------------------------------------------
P1       |  150             -50           -100
P2       |           50              50   -100
PL       | -100     -10                            110
PK       |  -50     -40                             90
PW       |                                 200    -200
PFX      |                   50     -50
----------------------------------------------------------------
$OFFTEXT

PARAMETERS
  F I X T     Fixed values of the tarifF
  O P T I M A L T Optimal tariff in the MPEC;

FIXT = 0;

NONNEGATIVE VARIABLES
   X 1      Activity level for sector X1,
   X 2      Activity level for sector X2,
   E 1      Activity level for sector E1,
   E 2      Activity level for sector E2,
   M 1      Activity level for sector M1,
   M 2      Activity level for sector M2,
   W       Activity level for sector W,
   P 1      Price index for commodity X,
   P 2      Price index for commodity Y,
   P L      Price index for primary factor L,
   P K      Price index for primary factor K,
   P W      Price index for welfare (expenditure function),
   P F X     Read exchange rate index,

   C O N S H   Income definition for home agent
   T O T     Terms of trade: world price of export good 1
   T M 2     Tariff - initially held fixed;

EQUATIONS
   P R F _ X 1  Zero profit for sector X1
   P R F _ X 2  Zero profit for sector X2
   P R F _ E 1  Zero profit for sector E1

   P R F _ E 2  Zero profit for sector E2
   P R F _ M 1  Zero profit for sector M1
   P R F _ M 2  Zero profit for sector M2
   P R F _ W   Zero profit for sector W

   M K T _ X 1  Supply-demand balance for commodity X1
   M K T _ X 2  Supply-demand balance for commodity X2
   M K T _ P F X Supply-demand balance for commodity PFX
   M K T _ L   Supply-demand balance for primary factor L
   M K T _ K   Supply-demand balance for primary factor L
   M K T _ W   Supply-demand balance for aggregate demand

   I _ C O N S H Income definition for CONSH
   T _ T O T   Equation for TOT
   T A R I F F  Dummy equation to fix the tariff as if a parameter;

*       Zero profit conditions

PRF_X1..   150*PL**(2/3) * PK**(1/3) =G= 150*P1;

PRF_X2..   50*PL**(1/4) * PK**(3/4)  =G= 50*P2;

PRF_E1..   50*P1 =G= 50*PFX*TOT;

PRF_E2..   50*P2 =G= 50*0.99*PFX;

PRF_M1..   50*PFX*TOT*1.01 =G= 50*P1;

PRF_M2..   50*PFX*(1+TM2) =G= 50*P2;

PRF_W..    200*P1**0.5*P2**0.5 =G= 200*PW;

*       Market clearance conditions

MKT_X1..   150*X1 + 50*M1 =G= 50*E1 + 100*W*PW/P1;

MKT_X2..   50*X2  + 50*M2 =G= 50*E2 + 100*W*PW/P2 ;

MKT_PFX..  50*0.99*E2 + (50*TOT)*E1 =G= 50*M2  + 50*1.01*M1;

MKT_W..    200*W =G= CONSH / PW;

MKT_L..    110  =G= 100*X1*P1/PL + 10*X2*P2/PL;

MKT_K..    90 =G= 50*X1*P1/PK + 40*X2*P2/PK;

*       Income balance, auxiliary equation

I_CONSH..  CONSH =E= 110*PL + 90*PK + 50*M2*PFX*TM2;

T_TOT..    TOT =G= E1**(-0.3);

TARIFF..   TM2 =E= FIXT;

MODEL LOE /PRF_X1.X1, PRF_X2.X2, PRF_E1.E1, PRF_E2.E2,
                 PRF_W.W, PRF_M1.M1, PRF_M2.M2,
                 MKT_X1.P1, MKT_X2.P2, MKT_PFX.PFX, MKT_L.PL,
                 MKT_K.PK, MKT_W.PW,
                 I_CONSH.CONSH, T_TOT.TOT, TARIFF.TM2 /;

*       Check the benchmark:

X1.L    =1;
X2.L    =1;
E2.L    =0;
E1.L    =1;
M2.L    =1;
M1.L    =0;
W.L     =1;
P1.L    =1;
P2.L    =1;
PFX.L   =1;
PK.L    =1;
PW.FX   =1;
PL.L    =1;
CONSH.L =200;
TOT.L   = 1;
TM2.L  = 0;

LOE.ITERLIM = 0;
SOLVE LOE USING MCP;

LOE.ITERLIM = 10000;
SOLVE LOE USING MCP;

*   Apply a tariff which improves the terms of trade and home
*   welfare:

FIXT  = 0.2;

SOLVE LOE USING MCP;

* now let's reformulate the problem as an MPEC to solve for the
* optimal tariff

VARIABLES
 WELOPT;

EQUATIONS
 WELFOPT;

WELFOPT..  WELOPT =E= W;

* ADD the objective function and DELETE the dummy equation TARIFF

MODEL OPTTARIFF /WELFOPT, PRF_X1.X1, PRF_X2.X2, PRF_E1.E1, PRF_E2.E2,
                 PRF_W.W, PRF_M1.M1, PRF_M2.M2,
                 MKT_X1.P1, MKT_X2.P2, MKT_PFX.PFX, MKT_L.PL,
                 MKT_K.PK, MKT_W.PW,
                 I_CONSH.CONSH, T_TOT.TOT/;

OPTION MPEC = nlpec;
SOLVE OPTTARIFF USING MPEC MAXIMIZING WELOPT;

OPTIMALT = TM2.L;
DISPLAY OPTIMALT;

* go back to the orignal MCP and "brute force" search for the optimal
* tariff and see if it matches the MPEC value

SETS I /I1*I51/;

PARAMETERS
 IMTARIFF(I)
 WELFARE(I);

LOOP(I,

FIXT = 0.005*ORD(I)**2 - 0.005;

SOLVE LOE USING MCP;

IMTARIFF(I) = TM2.L;
WELFARE(I) = W.L;

);

DISPLAY IMTARIFF, WELFARE;

The optimal tariff in this scenario is 0.480.  How does this compare to
the optimal tariff from economic theory?

Let t be the import tariff and let J (tao) be an export tax with the world
price as the base, world price denote with stars

Economic theory show that the optimal value of J should be given by
an inverse elasticity rule:  

J should equal the inverse of the foreign elasticity of demand
(defined as positive) for our export good.  

Let this be given by 0.   Then J = 1/0 .  Note that this is just saying
that marginal cost of exports (domestic price) should equal
marginal revenue.

marginal revenue on export sales

In our formulation above, 0.3 is the inverse elasticity, and hence the
optimum export tax is J = 0.3.

Does this hold in our numerical example?  The relationship between
an import tariff and export tax as defined above can be solved
from

In our example, t = 0.480 and so J = 0.324.  This is close to but not
equal to 0.30 so I am not quite sure what is going on.

Model M8-5b reformulates the problem with an export tax and arrives
at the same result: the optimal export tax is 0.324.

8.6a Two-country Heckscher-Ohlin model: Nash tariffs as an
iterative MPEC

This is the classic Heckscher-Ohlin model of trade: two goods, two
factors of production, and two countries.

     XHH  YHH  XHF  YHF  XFF  YFF  XFH  YFH   WH   WF   CONSH CONSF

PXH  150      -50                           -100
PYH        50                           50  -100
PXF            50         50                     -100
PYF                           150      -50       -100
PWH                                          200        -200
PWF                                               200         -200
PLH  -120 -10                                            130
PKH   -30 -40                                             70
PLF                      -40  -30                               70
PKF                      -10 -120                              130

Countries are symmetric, mirror images of each other.

$TITLE:  M8-6a.GMS: Full two-country Heckscher-Ohlin model
* formulation as an MPEC allows solutions for joint welfare max
* and non-cooperative Nash equilibrium in tariffs

$ONTEXT

     XHH  YHH  XHF  YHF  XFF  YFF  XFH  YFH   WH   WF   CONSH CONSF

PXH  150      -50                           -100
PYH        50                           50  -100
PXF            50         50                     -100
PYF                           150      -50       -100
PWH                                          200        -200
PWF                                               200         -200
PLH  -120 -10                                            130
PKH   -30 -40                                             70
PLF                      -40  -30                               70
PKF                      -10 -120                              130

$OFFTEXT

PARAMETERS
  SIZEH, S I Z E F  scales the endowments of countries h and f up and down
  WELBH, W E L B F  store free trade (B for benchmark) welfare level shares
  C A S E          denotes which case (of size differences) is displayed
  RESULTS(*, *), RESULTS2(*,*,*);

SIZEH = 1; SIZEF = 1;
WELBH = 0.5; WELBF = 0.5;

VARIABLES
  J W E L M A X       joint welfare maximization
  W E L H M A X       welfare of country h
  W E L F M A X       welfare of country f
  T A R H          tariff of country h
  T A R F          tariff of country f;

POSITIVE VARIABLES
  X H H     X home prododuction - home consumption
  Y H H     Y home prododuction - home consumption
  X H F     X home prododuction - foreign consumption
  Y H F     Y home prododuction - foreign consumption
  X F F     X foreign prododuction - foreign consumption
  Y F F     Y foreign prododuction - foreign consumption
  X F H     X foreign prododuction - home consumption
  Y F H     Y foreign prododuction - home consumption
  W H      Activity level for sector WH (Hicksian domestic welfare index)
  W F      Activity level for sector WF (Hicksian foreign welfare index)
  P X H     Price index for commodity home X
  P Y H     Price index for commodity home Y
  P X F     Price index for commodity foreign X
  P Y F     Price index for commodity foreign X

  P W H     Price index for welfare domestic (expenditure function)
  P W F     Price index for welfare foerign (expenditure function)
  P L H     home price labor
  P K H     home price capital
  P L F     foreign price labor
  P K F     foreign price capital
  C O N S H   Income definition for home agent
  C O N S F   Income definition for foreign agent;

EQUATIONS
  O B J J    objective function for maximizing joint welfare
  O B J H    objective function of country h (maximize WH)
  O B J F    objective function of country f (maximize WF)
  P R F _ X H H Zero profit for sector XHH
  P R F _ Y H H Zero profit for sector YHH
  P R F _ X H F Zero profit for sector XHF
  P R F _ Y H F Zero profit for sector YHF
  P R F _ X F F Zero profit for sector XFF
  P R F _ Y F F Zero profit for sector YFF
  P R F _ X F H Zero profit for sector XFH
  P R F _ Y F H Zero profit for sector YFH
  P R F _ W H  Zero profit for sector WH (Hicksian home welfare index)
  P R F _ W F  Zero profit for sector WF (Hicksian foreign welfare index)

  M K T _ X H  Supply-demand balance for commodity XH
  M K T _ X F  Supply-demand balance for commodity XF

  M K T _ Y H  Supply-demand balance for commodity YH
  M K T _ Y F  Supply-demand balance for commodity YF
  M K T _ L H  Supply-demand balance for primary factor LH
  M K T _ K H  Supply-demand balance for primary factor KH
  M K T _ L F  Supply-demand balance for primary factor LF
  M K T _ K F  Supply-demand balance for primary factor KF
  M K T _ W H  Supply-demand balance for aggregate home demand
  M K T _ W F  Supply-demand balance for aggregate foreign demand

  I _ C O N S F  Income definition for CONSF
  I _ C O N S H  Income definition for CONSH;

*       Zero profit conditions

OBJJ..       JWELMAX =E= WH**WELBH*WF**WELBF;
OBJH..       WELHMAX =E= WH;
OBJF..       WELFMAX =E= WF;

PRF_XHH..    150 * PLH**(4/5) * PKH**(1/5) =G= 150 * PXH;

PRF_YHH..    50  * PLH**(1/5) * PKH**(4/5) =G= 50 * PYH;

PRF_XFF..    50  * PLF**(4/5) * PKF**(1/5) =G= 50 * PXF;

PRF_YFF..    150 * PLF**(1/5) * PKF**(4/5) =G= 150 * PYF;

PRF_XHF..    50 * (1+TARF) * PXH =G= 50 * PXF;

PRF_YHF..    50.1 * PYH =G= 50 * PYF;

PRF_XFH..    50.1 * PXF =G= 50 * PXH;

PRF_YFH..    50 * (1+TARH) * PYF =G= 50 * PYH;

PRF_WH..     200 * PXH**0.5 * PYH**0.5 =G= 200 * PWH;

PRF_WF..     200 * PXF**0.5 * PYF**0.5 =G= 200 * PWF;

*       Market clearance conditions

MKT_XH..     150 * XHH + 50 * XFH =G= 50 * XHF  + 100 * WH * PWH /PXH;

MKT_XF..     50 * XFF + 50 * XHF =G= 50.01 * XFH + 100 * WF * PWF/PXF;

MKT_YH..     50 * YHH + 50 * YFH =G= 50.01 * YHF +100 * WH * PWH/PYH;

MKT_YF..     150 * YFF + 50 * YHF =G= 50 * YFH + 100 * WF * PWF/PYF;

MKT_LH..     130*SIZEH =G= 120 * XHH * PXH/PLH + 10 * YHH * PYH/PLH;

MKT_KH..     70*SIZEH  =G=  30 * XHH * PXH/PKH + 40 * YHH * PYH/PKH;

MKT_LF..     70*SIZEF  =G=  40 * XFF * PXF/PLF + 30 * YFF * PYF/PLF;

MKT_KF..     130*SIZEF =G= 120 * YFF * PYF/PKF + 10 * XFF * PXF/PKF;

MKT_WH..     200 * WH =G= CONSH/PWH;

MKT_WF..     200 * WF =G= CONSF/PWF;

*       Income balance states

I_CONSH..    CONSH =E= 130*SIZEH*PLH + 70*SIZEH*PKH + 50*YFH*PYF*TARH;

I_CONSF..    CONSF =E= 70*SIZEF*PLF + 130*SIZEF*PKF + 50*XHF*PXH*TARF;

OPTION MPEC=nlpec;

MODEL HO /
OBJJ,
OBJH,
OBJF,
PRF_XHH.XHH,
PRF_YHH.YHH,
PRF_XHF.XHF,
PRF_YHF.YHF,
PRF_XFF.XFF,
PRF_YFF.YFF,

PRF_XFH.XFH,
PRF_YFH.YFH,
PRF_WH.WH,
PRF_WF.WF,

MKT_XH.PXH,
MKT_XF.PXF,
MKT_YH.PYH,
MKT_YF.PYF,
MKT_LH.PLH,
MKT_KH.PKH,
MKT_LF.PLF,
MKT_KF.PKF,
MKT_WH.PWH,
MKT_WF.PWF,

I_CONSF.CONSF,
I_CONSH.CONSH
/;

*       Check the benchmark

XHH.L   =1;
YHH.L   =1;
XHF.L   =1;
YHF.L   =0;

XFF.L   =1;
YFF.L   =1;
XFH.L   =0;
YFH.L   =1;
WH.L    =1;
WF.L    =1;

PXH.L   =1;
PXF.L   =1;
PYH.L   =1;
PYF.L   =1;
PLH.L   =1;
PKH.L   =1;
PLF.L   =1;
PKF.L   =1;
PWH.FX  =1;
PWF.L   =1;

* "benchmark" with countries symmetric and free trade:
* all activities, prices = 1

HO.ITERLIM = 1000;
SOLVE HO USING MPEC MAXIMIZING JWELMAX;

* allow for size differences (scaling endowments up and down)

SIZEH = 1;
SIZEF = 1;

TARH.FX = 0;
TARF.FX = 0;

SOLVE HO USING MPEC MAXIMIZING JWELMAX;

* store free trade welfare levels, used to compute JWELMAX

WELBH = WH.L/(WH.L + WF.L);
WELBF = WF.L/(WH.L + WF.L);
DISPLAY WELBH, WELBF;

SOLVE HO USING MPEC MAXIMIZING JWELMAX;

RESULTS("SIZEH", "FREETR") = SIZEH;
RESULTS("WELJ", "FREETR") = JWELMAX.L;
RESULTS("WELH", "FREETR") = WH.L;
RESULTS("WELF", "FREETR") = WF.L;
RESULTS("TARIFFH", "FREETR") = TARH.L;
RESULTS("TARIFFF", "FREETR") = TARF.L;

* solve for the (unilateral by h) non-cooperative Nash equilibrium

TARH.LO = -INF;

TARH.UP = +INF;

SOLVE HO USING MPEC MAXIMIZING WELHMAX;

RESULTS("SIZEH", "UNIH") = SIZEH;
RESULTS("WELJ", "UNIH") = JWELMAX.L;
RESULTS("WELH", "UNIH") = WH.L;
RESULTS("WELF", "UNIH") = WF.L;
RESULTS("TARIFFH", "UNIH") = TARH.L;
RESULTS("TARIFFF", "UNIH") = TARF.L;

* solve for the joint welfare max

TARH.LO = -INF;
TARH.UP = +INF;
TARF.LO = -INF;
TARF.UP = +INF;

SOLVE HO USING MPEC MAXIMIZING JWELMAX;

RESULTS("SIZEH", "JMAX") = SIZEH;
RESULTS("WELJ", "JMAX") = JWELMAX.L;
RESULTS("WELH", "JMAX") = WH.L;
RESULTS("WELF", "JMAX") = WF.L;
RESULTS("TARIFFH", "JMAX") = TARH.L;
RESULTS("TARIFFF", "JMAX") = TARF.L;

DISPLAY RESULTS;

* now let's look at a non-cooperative outcome in tariffs
* iterative procedure:
*  max WH subject to TARF fixed
*  hold TARH at it's solution level and free up TARF
*  max WF solve model for fixed TARF
*  repeat 10 time

SETS J size difference for the countries /J1*J9/;
SETS I iterative procedure to determine Nash tariffs /I1*I10/;

LOOP(J,
 SIZEH = 0.9 + 0.1*ORD(J);
 SIZEF = 1.1 - 0.1*ORD(J);

TARH.FX = 0;
TARF.FX = 0;

SOLVE HO USING MPEC MAXIMIZING JWELMAX;

* store free trade welfare levels in order to get
* correct value of JWELMAX

WELBH = WH.L/(WH.L + WF.L);

WELBF = WF.L/(WH.L + WF.L);
SOLVE HO USING MPEC MAXIMIZING JWELMAX;

RESULTS2(J, "SIZEH", "FREETR") = SIZEH;
RESULTS2(J, "WELJ", "FREETR") = JWELMAX.L;
RESULTS2(J, "WELH", "FREETR") = WH.L;
RESULTS2(J, "WELF", "FREETR") = WF.L;
RESULTS2(J, "TARIFFH", "FREETR") = TARH.L;
RESULTS2(J, "TARIFFF", "FREETR") = TARF.L;

* loop to compute non-cooperative Nash equilibrium in tariffs
* fist solve for TARH given TARF, then TARF given the existing
* value of TARH.  Loop 10 times to get best-response Nash eq.

LOOP(I,

TARH.LO = -INF;
TARH.UP = +INF;
TARF.FX = TARF.L;

SOLVE HO USING MPEC MAXIMIZING WELHMAX;

TARF.LO = -INF;
TARF.UP = +INF;
TARH.FX = TARH.L;

SOLVE HO USING MPEC MAXIMIZING WELFMAX;

);

RESULTS2(J, "SIZEH", "NONCOOP") = SIZEH;
RESULTS2(J, "WELJ", "NONCOOP") = JWELMAX.L;
RESULTS2(J, "WELH", "NONCOOP") = WH.L;
RESULTS2(J, "WELF", "NONCOOP") = WF.L;
RESULTS2(J, "TARIFFH", "NONCOOP") = TARH.L;
RESULTS2(J, "TARIFFF", "NONCOOP") = TARF.L;

);

DISPLAY RESULTS, RESULTS2;

8.6b Two-country Heckscher-Ohlin model: Nash tariffs as an
iterative MPEC, using set notation

This is the same model, but illustrating the use of sets in problems
like this where countries, goods and factors can be entered more
compactly as sets.

SETS I goods     /1, 2/
     J factors   /1, 2/
     C countries /H, F/;

Data is read in as a table, and then values such as representative
quantities and Cobb-Douglas share parameters in production and
utility are calculated.

Not that no number appear in the equations of the model.

$TITLE  M8-6b.GMS: 2x2x2 of M8-6a in set notation

TABLE BENCH(*,*)
     XHH  YHH  XHF  YHF  XFF  YFF  XFH  YFH   UH   UF   CONSH CONSF
PXH  150      -50                           -100
PYH        50                           50  -100
PXF            50         50                     -100
PYF                           150      -50       -100
PUH                                          200        -200
PUF                                               200         -200
PLH  -120 -10                                            130
PKH   -30 -40                                             70
PLF                      -40  -30                               70
PKF                      -10 -120                              130;

DISPLAY BENCH;

SETS I g o o d s     /1, 2/
     J f a c t o r s   /1, 2/
     C c o u n t r i e s /H, F/;

ALIAS (J,JJ), (I,II);

PARAMETERS

 VS(J,C)       endowment of factor j in country c

 TC(I)         trade cost of importing good i for country c
 BETA(I,J)     share of factor j in the production of good i
 GAMMA(I)      share of good i in the utility function
 WELW(C)       welfare weight of country c in world welfare
 RESULTS(*,*)  assemble the results;

VS("1", "H") = BENCH("PLH", "CONSH");
VS("2", "H") = BENCH("PKH", "CONSH");
VS("1", "F") = BENCH("PLF", "CONSF");
VS("2", "F") = BENCH("PKF", "CONSF");

BETA("1","1") = -BENCH("PLH", "XHH")/ BENCH("PXH", "XHH");
BETA("1","2") = -BENCH("PKH", "XHH")/ BENCH("PXH", "XHH");
BETA("2","1") = -BENCH("PLH", "YHH")/ BENCH("PYH", "YHH");
BETA("2","2") = -BENCH("PKH", "YHH")/ BENCH("PYH", "YHH");

GAMMA("1")    = -BENCH("PXH", "UH")/ BENCH("PUH", "UH");
GAMMA("2")    = -BENCH("PYH", "UH")/ BENCH("PUH", "UH");

WELW("H") = BENCH("PUH", "UH")/(BENCH("PUH", "UH")+BENCH("PUF", "UF"));
WELW("F") = BENCH("PUF", "UF")/(BENCH("PUH", "UH")+BENCH("PUF", "UF"));

DISPLAY VS, BETA, GAMMA, WELW;

TC(I)    = 1.0001;

VARIABLES
  J W E L M A X        joint welfare maximization
  W E L M A X H        welfare of country H
  W E L M A X F        welfare of country F
  TR(I,C)        tariff of country C on good I;

POSITIVE VARIABLES
        X(I,C)   Activity level for prduction of I by C
        M(I,C)   Activity level for imports of I by C
        E(I,C)   Activity level for exports of I by C
        U(C)     Activity level for sector U in country C

        PX(I,C)  Price of commodity I in country C
        PW(I)    World price of commodity I
        PV(J,C)  Price of factor J in country C
        PU(C)    Price of welfare (expenditure function) in country C

        CONS(C)  Aggregate income;

EQUATIONS
        OBJJ
        OBJH, OBJF
        PRX(I,C)   Zero profit for sector I in country J
        PRM(I,C)   Zero profit for imports of I by country j
        PRE(I,C)   Zero profit for exports of I by country j
        PRU(C)     Zero profit for sector U

        MKX(I,C)   Supply-demand balance for commodity I
        MKW(I)     Supply-demand balance for imports and exports of I
        MKV(J,C)   Supply-demand balance for primary factor V1
        MKU(C)     Supply-demand balance for welfare (aggregate demand)

        ICONS(C)  Income definition for CONS;

OBJJ..      JWELMAX =E= PROD(C, U(C)**WELW(C));

OBJH..      WELMAXH =E= U("H");
OBJF..      WELMAXF =E= U("F");

*       Zero profit inequalities

PRX(I,C)..  PROD(J, PV(J,C)**BETA(I,J))  =G=  PX(I,C);

PRM(I,C)..  PW(I)*TC(I)*TR(I,C)  =G=  PX(I,C);

PRE(I,C)..  PX(I,C)  =G=  PW(I);

PRU(C)..    PROD(I, PX(I,C)**GAMMA(I)) =G=  PU(C);

*       Market clearance inequalities

MKX(I,C)..  X(I,C)- E(I,C) + M(I,C)/TC(I) =G=

            PROD(II, PX(II,C)**GAMMA(II))*(GAMMA(I)/PX(I,C))*U(C);

MKW(I)..    SUM(C, E(I,C) - M(I,C)) =G= 0;

MKV(J,C)..  VS(J,C)  =G=  SUM(I,
            PROD(JJ, PV(JJ,C)**BETA(I,JJ))*(BETA(I,J)/PV(J,C))*X(I,C));

MKU(C)..    U(C) =E= CONS(C) / PU(C);

*       Income balance equations

ICONS(C)..  CONS(C) =E= SUM(J, VS(J,C)*PV(J,C)) +
                        SUM(I, PW(I)*(TR(I,C) - 1)*M(I,C));

MODEL MCP /PRX.X, PRM.M, PRE.E, PRU.U,
                 MKX.PX, MKW.PW, MKV.PV, MKU.PU,
                 ICONS.CONS/;

MODEL MPEC /OBJJ, OBJH, OBJF, PRX.X, PRM.M, PRE.E, PRU.U,
                 MKX.PX, MKW.PW, MKV.PV, MKU.PU,
                 ICONS.CONS/;

PU.L(C) = 1;
PU.FX("H") = 1;

*       Set initial values of variables:

X.L(I,C)       =100;
M.L(I,C)       =100;
E.L(I,C)       =100;
U.L(C)         =200;
PX.L(I,C)      =1;
PW.L(I)        =1;
PV.L(J,C)      =1;
PV.L(J,C)      =1;
CONS.L(C)      =200;
WELMAXH.L      =U.L("H");
WELMAXF.L      =U.L("F");
JWELMAX.L      = 1;

* fix tariffs at zero (meaning TR = 1) to solve for free trade

TR.FX(I,C) = 1;

OPTION MPEC = nlpec;
SOLVE MPEC USING MPEC MAXMIZING JWELMAX;

RESULTS("WELJ", "FREETR") = JWELMAX.L/200;
RESULTS("WELH", "FREETR") = U.L("H")/200;
RESULTS("WELF", "FREETR") = U.L("F")/200;
RESULTS("TARIFFH", "FREETR") = TR.L("2","H")-1;
RESULTS("TARIFFF", "FREETR") = TR.L("1","F")-1;

DISPLAY RESULTS;

* show that this can be done with the mcp version when TR variables
* are fixed

SOLVE MCP USING MCP;
TR.FX(I,"H") = 1.2;
SOLVE MCP USING MCP;

* solve for the optimal tariff for country h when TRF is still fixed
* at zero.

TR.UP("2", "H") = +INF;
TR.LO("2", "H") = -INF;

SOLVE MPEC USING MPEC MAXIMIZING WELMAXH;

RESULTS("WELJ", "UNIH") = JWELMAX.L/200;
RESULTS("WELH", "UNIH") = U.L("H")/200;
RESULTS("WELF", "UNIH") = U.L("F")/200;
RESULTS("TARIFFH", "UNIH") = TR.L("2","H")-1;
RESULTS("TARIFFF", "UNIH") = TR.L("1","F")-1;
DISPLAY RESULTS;

*$EXIT

SETS K iterative procedure to determine Nash tariffs /K1*K10/;

TR.FX(I,C) = 1;

LOOP(K,

TR.UP("2", "H") = +INF;
TR.LO("2", "H") = -INF;
TR.FX("1", "F") = TR.L("1", "F");

SOLVE MPEC USING MPEC MAXIMIZING WELMAXH;

TR.UP("1", "F") = +INF;
TR.LO("1", "F") = -INF;
TR.FX("2", "H") = TR.L("2", "H");

SOLVE MPEC USING MPEC MAXIMIZING WELMAXF;

);

RESULTS("WELJ", "NASH") = JWELMAX.L/200;
RESULTS("WELH", "NASH") = U.L("H")/200;
RESULTS("WELF", "NASH") = U.L("F")/200;
RESULTS("TARIFFH", "NASH") = TR.L("2","H")-1;
RESULTS("TARIFFF", "NASH") = TR.L("1","F")-1;
DISPLAY RESULTS;

$LIBINCLUDE XLDUMP RESULTS M8.XLS SHEET2!A3




